This tutorial demonstrates the behaviour of a two-level system upon action of periodic perturbation. Just evaluate the notebook. Then use the last input cell and change the parameters for your enjoyment!

S. Rochester and D. Budker; October 6, 1999

Here we define the Schroedinger Equation (SE) and the wave function in the 2-state basis:

```
SE[H_, Ψ_] := ⅈ ∂_t Ψ == H.Ψ
```

```
MatrixForm[Ψ[t_] = {a[t], b[t]}]
```

$$\begin{pmatrix} a[t] \\ b[t] \end{pmatrix}$$

Explicit form of the Hamiltonian and the SE;
note that the Hamiltonian is not quite Hermitian because of the $i\,\Gamma/2$:

$$H = \begin{pmatrix} 0 & v0\, e^{i\,\omega\,t} \\ v0\, e^{-i\,\omega\,t} & \omega 0 - i\,\Gamma/2 \end{pmatrix};$$

```
MatrixForm[eqs = {LogicalExpand[SE[H, Ψ[t]]][[1]], LogicalExpand[SE[H, Ψ[t]]][[2]]}]
```

$$\begin{pmatrix} i\, a'[t] == e^{i\,t\,\omega}\, v0\, b[t] \\ i\, b'[t] == e^{-i\,t\,\omega}\, v0\, a[t] + \left(-\frac{i\,\Gamma}{2} + \omega 0\right) b[t] \end{pmatrix}$$

Initial condition: all atoms are in the ground state at t=0

```
MatrixForm[Ψ0 = {1, 0}]
inits = {LogicalExpand[Ψ[0] == Ψ0][[1]], LogicalExpand[Ψ[0] == Ψ0][[2]]};
```

```
General::spell1 : Possible spelling error: new symbol name "Ψ0" is similar to existing symbol "ω0".
```

$$\begin{pmatrix} 1 \\ 0 \end{pmatrix}$$

Now we solve the SE explicitly. Here is the place where we put numerical values for various parameters. Start with Γ=0, ω=ω0.

```
sols = NDSolve[Union[eqs /. {Γ → 0, ω0 → 20, ω → 20, v0 → 1}, inits],
    {a, b}, {t, 0, 100}, MaxSteps → 10000][[1]];
```

Now we plot the solution. Here we show the probability of finding the system in the excited state. We see the Rabi oscillations. Note that the oscillation (Rabi) frequency is Ω_R =2v0.

```
Plot[{Abs[b[t]]^2 /. sols}, {t, 0, 10}, PlotRange → {0, 1}, AxesLabel → {t, Probability}]
```




```
- Graphics -
```

Now let us "turn on" relaxation. First let's try Γ<<v0:

```
sols = NDSolve[Union[eqs /. {Γ → .3, ω0 → 20, ω → 20, v0 → 1}, inits],
    {a, b}, {t, 0, 100}, MaxSteps → 10000]⟦1⟧;
Plot[{Abs[b[t]]^2 /. sols}, {t, 0, 10}, PlotRange → {0, 1}, AxesLabel → {t, Probability}]
```



- Graphics -

We see damped oscillation. Now try Γ>>v0:

```
sols = NDSolve[Union[eqs /. {Γ → 3, ω0 → 20, ω → 20, v0 → .3}, inits],
    {a, b}, {t, 0, 100}, MaxSteps → 10000]⟦1⟧;
Plot[{Abs[b[t]]^2 /. sols}, {t, 0, 10}, PlotRange → {0, .1}, AxesLabel → {t, Probability}]
```



- Graphics -

Now we see that there are no more Rabi oscillations. This the "overdamped" regime. Note that at first the upper state population grows as if there was no relaxation, but then it "saturates" at a small level P_{max}~$\left(\frac{2\,v0}{\Gamma}\right)^2$, and then eventually decays away. The maximum upper state population occurs at a time t_{max}~$2\pi/\Gamma$.

More things to try and think about:

1. What happens if the frequency ω is different from the

resonance frequency $\omega 0$?

2. For $\omega=\omega 0$, what is the critical value of $v0/\Gamma$ at which the "overdamped" regime turns into oscillatory regime?

3. In the above, we assumed that the relaxation occurs due to the decay into unobserved states. What will change if the excited state decays back to the ground state?